

A BASIC Printing Program For Floppy Labels

THIS BASIC PROGRAM AND BATCH FILE PRINT A FLOPPY DISK'S CONTENTS ON A LABEL. **BY WILLIAM BARDEN, JR.**

EMBARRASSED TO HAND out disks with handwritten labels? Tired of fiddling with typewriters or word processing macros to produce printed disk labels? A short program called DLABEL prints the contents of a 3½-inch or 5¼-inch disk onto any size label.

From the DOS prompt, enter a command with five parameters, and DLABEL generates a label that lists every file on the disk alphabetically. DLABEL lets you specify which label on the sheet is to be printed, allowing you to move around on a sheet of labels until the last one has been printed. You can also enter

a title line on the label, which is centered automatically on the printout. DLABEL prints in condensed 7.5-point type with minimal line spacing to fit as many filenames on a label as possible. The output is arranged like this, with filenames separated by hyphens.

Utilities Disk

ASCXEBC.BAT-ASCXEBCP.BAS-ASCXEBCP.EXE-BATCH.BAS-CBLD.FOR-CMED.BLD-COLORS.BAS-COLORS.BAT-SWCRUNCH.BAS-WHEREIS.COM

To generate and use this program, you need the following elements: either

GW-Basic or QBasic (which ship with various versions of DOS), a text editor, a supply of labels, and either a LaserJet-compatible laser printer or an Epson/IBM-compatible dot matrix printer.

DLABEL.BAS and DLABEL.BAT

The DLABEL system consists of a BASIC program and a short batch file. Because of the differences in printer commands, there are two variations of DLABEL.BAS—one for LaserJet-compatible laser printers, and one for Epson/IBM-compatible dot matrix printers.

You can download both of these programs from PC/Contact, *PC/Computing's* online service, (see page 270 for details). The LaserJet-compatible version is named DLABEL.EXE; the dot matrix version is named DLABEL.DOTMEXE. Both are in data library 1 (Hot Tips/Secrets). Each program comes with the required batch file.

You can also create the appropriate version of the program and batch file using a text editor such as DOS 5.0's EDIT, a BASIC editor, or any word processing program that generates ASCII text. Both files must consist only of printable, displayable characters—no word processing control codes or characters other than letters, numbers, and punctuation.

If you have a LaserJet-compatible printer (most laser printers conform to Hewlett-Packard's Printer Control Language codes), enter the lines as shown in the program listing of DLABEL.BAS. If you have an Epson/IBM-compatible dot matrix printer (most dot matrix printers are Epson-compatible) enter the lines of the DLABEL.DOTMEXE listing, but substitute the following lines in place of lines 270 and 280:

```
270 PRINT#3,CHR$(27)+CHR$(80)+CHR$(15);
280 PRINT#3,CHR$(27)+CHR$(51)+CHR$(20);
    FOR M=1 TO YPOS: PRINT#3,CRLF$;
    NEXT
```

Save the file as DLABEL.BAS in either your batch or utilities subdirectory, and then open a new file. Copy the lines shown in the listing for the batch file DLABEL.BAT. If your version of DOS is earlier than 5.0, you probably have the GW-Basic compiler instead of QBasic. If this is the case, substitute the following line for the next-to-last line of the listing:

DLABEL.BAS

```
100 ' LABEL PRINTER FOR LASERJET COMPATIBLE
110 ON ERROR GOTO 360
120 DIM N$(1000): WIDTH "lpt1:",255: OPEN "lpt1:" AS #3: CRLF$=CHR$(10)+CHR$(13)
130 L$=ENVIRON$("LHEAD"):P$=ENVIRON$("LPATH"):WID=VAL(ENVIRON$("LWID"))*16.67
140 I=INSTR(L$,"."):IF I<>0 THEN MID$(L$,I)=" ": GOTO 140
150 XPOS=VAL(ENVIRON$("LIN"))*16.67: YPOS=VAL(ENVIRON$("LDOWN"))*9
160 SHELL "dir " + P$ + " >delete.me": OPEN "i",#1,"delete.me"
170 OPEN "o",#2,"delete.me2":WHILE NOT EOF(1): LINE INPUT#1,A$
180 IF A$="" OR LEFT$(A$,1)=" " OR LEFT$(A$,1)="." THEN GOTO 240
190 P=INSTR(A$," "): B$=LEFT$(A$,P-1)
200 IF MID$(A$,10,1)=" " THEN C$="" ELSE C$="."+MID$(A$,10,3)
210 IF MID$(A$,14,1)="<" THEN C$="<DIR>"
220 P=INSTR(C$," "):IF P<>0 THEN C$=LEFT$(C$,P-1)
230 PRINT#2,B$+C$
240 WEND: CLOSE 1,2: SHELL "sort <delete.me2 >delete.me":OPEN"i",#1,"delete.me"
250 I=0: WHILE NOT EOF(1): LINE INPUT#1, N$(I): I=I+1: WEND: CLOSE 1,2
260 KILL "delete.me2": KILL "delete.me": J=0: L$=STRING$(WID-LEN(L$)/2," ") + L$
270 PRINT#3,CHR$(27)+"(10U"+CHR$(27)+"(s0p16.67h8.5v0s0b0T"+CHR$(27)+"&100"
280 PRINT#3,CHR$(27)+"&15.33C": FOR M=1 TO YPOS: PRINT#3,CRLF$: NEXT
290 GOSUB 370: PRINT#3,L$+CRLF$: COL=0: GOSUB 370
300 FOR K=0 TO I-1:IF (COL+1+LEN(N$(K)))<=WID THEN GOTO 320
310 PRINT#3,CRLF$:GOSUB 370: COL=0:LN=LN+1: GOTO 340
320 IF COL<>0 THEN PRINT#3,"-"+N$(K): COL=COL+ LEN(N$(K))+1: GOTO 340
330 PRINT#3,N$(K): COL=COL+LEN(N$(K))
340 IF COL=0 THEN K=K-1
350 NEXT: PRINT#3,CHR$(12)
360 CLOSE: SYSTEM
370 PRINT#3,STRING$(XPOS," "): RETURN
```

PCContact

GW BASIC DLABEL.BAS

Save the file as DLABEL.BAT in your batch or utilities subdirectory, then quit your text editor. Verify that the batch directory is in your DOS search path by typing PATH and checking the listing. Also, check that your BASIC executable file is in your DOS directory. You're now ready to test the program.

Running DLABEL

The trickiest element of using DLABEL is aligning the labels and measurement parameters, so you might want to start with regular paper stock until you've worked out the fine points. Enter DLABEL at the DOS prompt, followed by four parameters separated by one or more blanks. Here's an example:

```
dlabel This.is.a.Title a: 1 2 3
```

The first parameter after DLABEL is a title for the label. Use any short text

DLABEL.BAT

```
@echo off
cls
if not "%1"==" " goto next
echo DLABEL prints disk labels using the following command-line parameters.
echo DLABEL [Title] [Drive:] [LeftMargin] [TopMargin] [Width]"
echo "Title" can be any text: ALWAYS use periods instead of spaces between words;
echo "drive:" can be a:, b:, or any drive notation or directory path;
echo "leftmargin" is the label's offset in inches from left
echo "topmargin" is the label's offset in inches from top
echo "width" is the label's width in inches.
goto out
:next
if exist %2*. * goto next1
echo %2 does not exist!
goto out
:next1
set lhead=%1
set lpath=%2
set lin=%3
set ldown=%4
set lwid=%5
qbasic /run dlabel.bas
:out
```

PCContact



HOW TO UNDERSTAND SHAREWARE.

"PC/Computing Guide to Shareware" is the definitive shareware reference. In it, Preston Gralla of *PC/Computing* gives you his in-depth, authoritative evaluation of the 250 programs he considers the best in the shareware market. You'll learn about their strengths, weaknesses, affordable pricing and how to get them. And as a bonus, you'll also receive two free disks filled with the hottest programs available. Call **1-800-688-0448** to order the "PC/Computing Guide to Shareware" today (for \$34.95 suggested retail price). Simply ask for ISBN 1-56276-036-X.



string with any printable characters for a title, but don't use spaces. For a title that contains more than one word, just insert periods instead of spaces between the words. The program translates the periods into spaces when the title is printed. It also centers the title automatically.

The second parameter is the letter of the drive whose contents you want to list. In this example, the drive is A:, but you can have B: or any other drive, even hard drives. You can also use the path to a subdirectory for a label that lists only the subdirectory contents. (For example, specify A:\FEBRUARY to list the contents of the FEBRUARY subdirectory on floppy drive A:.)

The third, fourth, and fifth parameters allow you to fine-tune your labels. They are measurements in inches for, respectively, the left margin, the top margin, and the width of the label. In the example above, the printing occurs in a


box one inch from the left edge of the paper and two inches from the top, and on a label three inches wide. But bear in mind that the measurements the program uses aren't precise—they are accurate to within one-ninth of an inch vertically by one-seventeenth of an inch horizontally. The exact position of the print box on a sheet of labels also depends, of course, on the position of the paper on a dot matrix print platen or the margins on a laser printer. However, you can get accurate placement by trial and error. Before printing labels, run a plain-paper test to determine the input parameters. Keep the measurements of the size label you use on hand for reference.

A 5¼-inch disk has space for 1¾-by-5-inch labels, while a 3½-inch disk allows for 2-by-2¾-inch labels. You can get precut labels in a variety of shapes from numerous sources, such as Avery International. Another option is to buy an 8½-

by-11-inch label sheet; you can print and cut a label, then reuse additional portions of the sheet by specifying new dimensions to DLABEL.

DLABEL prints at about 16.7 characters per inch horizontally by 9 lines per inch vertically. You can print about 10 lines including the title, or approximately 65 files, on a 5¼-inch label. The program does *not* check for overrun on the number of lines—it continues printing until all the filenames are listed, even if this requires 20 lines. If many of your disks contain more than 65 files, you might want to go to a larger label size. Another option for skilled BASIC programmers with a wide range of fonts: Go to a smaller font and smaller line spacing by altering lines 270, 280, and the 9 multiplier in line 150. ■

William Barden, Jr., is a PC consultant and the author of more than 30 computer-related books.



How To GET IT.

Every program in the "PC/Computing Guide to Shareware" is available through ZiffNet, the on-line service all about computing. You can easily download programs from a fully indexed software library.* ZiffNet also plugs you into information on what to buy, delivers industry news even before it's in print, and offers forums hosted by the editors of *PC Magazine* and *PC/Computing* to ask for advice or discuss issues. **Call 1-800-666-0330.** And remember, if you want to get into shareware, you need just two things. This guide. And ZiffNet.

Ziff NET.

*CompuServe and ZiffNet members type GO ZNT:PBS for direct access to the software library or type GO ZIFFNET to reach the ZiffNet main menu. © 1992 Ziff Communications.

Circle 231 on reader service card.